

/PAL III REVISED AS OF
/15 NOVEMBER 1966

/PAGE 1

/PAGE ZERO CONSTANTS AND STORAGE

/INCREASE TIME-OUT DELAY TO ACCOMMODATE PHOTO-ELECTRIC READER

*1

WAIT, 0

TAD M2

DCA 14

ISZ 14

JMP , -1

ISZ WAIT

JMP I WAIT

/INDIRECT REFERENCE TABLE

*20

AAA,	0	/POINTER TO READ ROUTINE
BBB,	0	/POINTER TO PUNCH ROUTINE
DDD,	0	/POINTER TO TYPE ROUTINE
SPECI,	SPSHUL	/SPECIAL CHARACTER HANDLER
NCKEKI,	NCKEK	/CHARACTER TYPE PROCESSOR
CHEKI,	CHEK	/CHECKS CHARACTER LEGALITY
SYMCMI,	SYMCOM	/SYMBOL COMPARISON ROUTINE
PUNI,	BINP	/BINARY PUNCH ROUTINE
EEE,	ALFO	/ALPHA OUTPUT ROUTINE
OCTI,	OCTO	/OCTAL OUTPUT ROUTINE
NUMBI,	NMBR	/OCTAL NUMBER PROCESSOR
SPACI,	SPACE	/SPACE HANDLER
CARRI,	CARR	/CARRIAGE RETURN HANDLER
DOLLI,	DOLLAR	/END-OF-PASS ROUTINE
STARI,	STAR	/ORIGIN HANDLER
ADDI,	ADD	/SET SIGN FOR ADD
SYMSTI,	SYMSTO	/COMMA HANDLER
SUBI,	SUB	/SET SIGN FOR SUBTRACT
POINTI,	POINT	/PERIOD PROCESSOR
EQUALI,	EQUAL	/PARAMETER ASSIGNMENTS
FFF,	CSYM	/MAIN EXPRESSION HANDLER
PUSHJI,	PUSHJM	/PUSHJ ROUTINE
POPJMI,	POPJMP	/POPJ ROUTINE
PUSHI,	PUSHIN	/PUSH ROUTINE
POPMI,	POPM	/POP ROUTINE
LDTI,	LQTR	/LEADER-TRAILER GENERATOR
LOPUNI,	LOPUN	/ASR 33 PUNCH
HIPUNI,	HIPUN	/75 A PUNCH
LOREDI,	LOREAD	/ASR 33 READER
HIREDI,	HIREAD	/750 READER
IORIN,	IOR	/POINTER TO INCLUSIVE OR ROUTINE
NEXI,	NXPA	/POINTER TO NEXT PASS

/PAGE 2

/NUMERICAL CONSTANTS

C2, 0002
C3, 0003
C4, 0004
C5, 0005
C6, 0006
C200, 0200
C211, 0211
C214, 0214
C215, 0215
C377, 0377
C4TH, 4000

M2, -0002
M3, -0003
M12, -0012
M23, -0023
M4, -0004
M17, -0017
M215, -0215
M200, -0200
M1, -0001
M6, -0006
M211, -0211
M212, -0212

/TEMPORARY REGISTERS

ITEM, 0	/USED BY INCLUSIVE OR ROUTINE
TEMS, 0	/USED BY EQUAL ROUTINE
TEM1, 0	/AVAILABLE TO ALL ROUTINES
TEM2, 0	/...
TEM3, 0	/...
POPTM, 0	/USED BY PUSH,POP,ETC

/PAGE 3

/SYMBOLIC CONSTANTS AND POINTERS

IAM1,	SYTA-1	/USED FOR AUTO-INDEX REGISTERS
PAGE,	7600	/MASK FOR PAGE BITS
CRLF,	4543	/CONSTANT FOR CR,LF
MASK,	3777	/MASKS OFF UNDEFINED BIT
SL3,	0007	/MASK FOR THREE BIT OCTAL #
SL7,	0177	/MASK FOR SEVEN BIT ADDRESS
SL6,	0077	/MASK FOR 6-BIT ALPHA CHAR
SF2,	6000	/MASK OFF PASS BITS
RBGN,	0	/BEGINNING OF READER BUFFER
RKON,	-574	/SIZE OF READER BUFFER
P1BGN,	PSHLST	/BEGINNING OF PUSHDOWN LIST
P1END,	-PSHLST-10	/CONSTANT FOR PDL OVERFLOW
TBUF,	7450	/BEGINNING OF PASS 3 BUFFER
GOREAD,	6650	/BUFFER WITH HIGH SPEED READER
C400,	0400	/INDIRECT BIT

/PAGE 4

/DATA REGISTERS

PF,	0	/PASS FLAG
SPNT,	0	/POINTER TO END OF SYMBOL TABLE
CHAR,	0	/CURRENT CHARACTER BEING PROCESSED
INFO,	0	/INFORMATION FLAG
BINW,	0	/12 BINARY ASSEMBLED WORD
OCWD,	0	/12 BIT OCTAL WORD
AACT,	0	/ABSOLUTE ADDRESS COUNTER
ADDR,	0	/12 BIT ASSEMBLED ADDRESS
SYMVAL,	0	/OCTAL VALUE OF SYMBOL
SYM1,	0	/FIRST TWO CHARACTERS OF 6-LETTER SYMBOL
SYM2,	0	/2ND TWO CHARACTERS
SYM3,	0	/3RD TWO CHARACTERS
CKSM,	0	/CURRENT BINARY CHECKSUM
INSF,	0	/INSTRUCTION FLAG; 1 IF INSTRUCTION
ZFLG,	0	/Z FLAG; 1 IF PAGE ZERO REFERENCE
FLAG,	0	/COMMENT FLAG; 1 IF IN A COMMENT
SIGN,	0	/SIGN ASSOCIATED WITH PRESENT ATOM
TYPE,	0	/SYMBOL TYPE
SWITCH,	0	/END-OF-TAPE FLAG
RCNT,	0	/NUMBER OF CHARACTERS LEFT IN BUFFER
PSHPN1,	0	/PUSH DOWN POINTER

/75A PUNCH ROUTINE

/LENGTH: 6 REGISTERS

HIPUN,	0	
PSF		
JMP	.-1	
PLS		
CLA		/EXIT WITH C(AC)=0
JMP	I HIPUN	

/PAGE 5
/PARAMETER ASSIGNMENTS

PUSHJ=JMS I PUSHJI
POPJ=JMP I POPJMI

AUTOA=10
AUTOB=11
AUTOC=13

SORT1=16
SORT2=17

POP=JMS I POPMI
PUSH=JMS I PUSHI

/DECIMAL AND OCTAL PSEUDO-OPS
/TO SET CURRENT RADIX
/LENGTH : 8 REGISTERS

DECMA,	TAD INST1	/NOP INSTRUCTION
	DCA I SWTCHN	/STORE IN SWITCH
	POPJ	
OCTALP,	TAD INST2	/SKP INSTRUCTION
	JMP .-3	/STORE AND EXIT
INST1,	NOP	
INST2,	SKP	
SWTCHN,	NSWTCH	/LOCATION IN "NMBR"
C204,	0204	
L00,		/END OF PAGE MARKER

/PAGE 6
 /BEGINNING OF PAL
 /DETERMINES TYPE OF I-O TO BE USED
 /CALCULATES PASS NUMBER FROM SWITCH SETTINGS
 /LENGTH: 35 REGISTERS
 *200

SPAL,	DCA TEM1	/0 TO COUNTER
	PLS	/SELECT PUNCH
	RFC	/SELECT READER
	RSF	/READER FLAG SET?
	JMS WAIT	
	JMP HREAD	/YES: HIGH READER INPUT
	ISZ TEM1	/WAITED ENOUGH?
	JMP .-4	/NO: TRY AGAIN
	TAD LOREDI	/YES: LOW SPEED READER
	DCA AAA	
	TAD TBUF	/ALL OF CORE FOR SYMBOLS
	DCA RBGN	
TPUN,	DCA TEM1	/0 TO COUNTER
	PSF	/PUNCH FLAG SET YET?
	SKP	/NO
	JMP SHIPUN	/YES: HIGH SPEED PUNCH
	ISZ TEM1	/WAITED ENOUGH?
	JMP .-4	/NO: TRY AGAIN
	TAD LOPUNI	/YES: LOW SPEED PUNCH
	DCA BBB	
PASS,	KCC	/ININITIALIZE DEVICES
	TLS	
	PLS	
	CLA OSR	
	AND SF2	/WAS PASS NUMBER SET?
	SNA	
	JMP NXPA	/NO, HALT
	RAL CLL	/GET HIGH ORDER BIT
	SNL	/WAS IT SET?
	JMP PASS1	/NO, PASS 1
	SMA CLA	/YES, WHAT ABOUT LOW ORDER?
	JMP PASS2	/NOT ON, PASS 2
	JMP PASS3	/ON, PASS3
NXPA,	HLT	
	JMP PASS	

/PAGE 7
/HREAD; HIPUN
/ROUTINES TO SET POINTERS FOR READ AND PUNCH
/LENGTH: 6 REGISTERS

HREAD, TAD HIREDI
DCA AAA
TAD GOREAD
JMP TPUN-1
SHIPUN, TAD HIPUNI
JMP PASS-1
/POPJ ROUTINE TO EXIT RECURSIVELY FROM SUBROUTINES
/CHECKS FOR PUSHDOWN UNDERFLOW VIA POP ROUTINE
/LENGTH: 3 REGISTERS

POPJMP, POP /GET RETURN ADDRESS
DCA POPTM /STICK IT IN POPTM
JMP I POPTM /EXIT

/ROUTINES TO SET PASS FLAG AND APPROPRIATE POINTERS
/LENGTH: 13 REGISTERS

PASS1, CLA /SET PASS FLAG TO ZERO
DCA PF
TAD IAM1 /INITIALIZE SYMBOL TABLE POINTER
DCA SPNT
JMP INITAL /GO INITIALIZE FLAGS, STORAGE
PASS2, IAC /SET PASS FLAG TO ONE
DCA PF
DCA CKSM /CLEAR THE CHECKSUM
JMS I LDTI /PUNCH LEADER-TRAILER
JMP INITAL /INITIALIZE FLAGS AND STORAGE
PASS3, CMA /SET PASS FLAG TO -1
DCA PF
JMS I LDTI /PUNCH LEADER/TRAILER

/PAGE 8

/INITIAL ROUTINE

/ROUTINE TO INITIALIZE FLAGS, BUFFERS, AND STORAGE

/LENGTH: 15 REGISTERS

INITAL, DCA SWITCH	/SET END OF TAPE SWITCH
TAD TBUF	/SET PASS3 BUFFER
DCA AUTOC	
TAD P1BGN	/SET PUSHJ,POPJ BUFFER
DCA PSHPN1	
CMA	/SET HIGH SPEED READER BUFFER
DCA RCNT	
TAD C200	/SET ORIGIN TO 200
PUSHJ	/SET
STAR+3	/AND PUNCH IF PASS 2
TAD LOPUN1	/TYPE ON PASS 3
DCA DDD	
DCA FLAG	/COMMENT SWITCH
PUSHJ	/SET RADIX TO OCTAL
OCTALP	

/PAGE 9
 /MAIN ROUTINE
 /MAIN CALLS CSYM TO ASSEMBLE A LINE OF CODE
 /AND DECIDES IF ANY PUNCHING OR PRINTING IS TO BE DONE
 /AND WHETHER THE AACT SHOULD BE INCREMENTED
 /LENGTH: 27 REGISTERS

MAIN,	DCA INSF	/CLEAR INSTRUCTION FLAG
	DCA ADDR	/CLEAR 12-BIT ADDRESS WORD
	DCA BINW	/CLEAR BINARY WORD
	DCA OCWD	/CLEAR OCTAL WORD
	DCA SYMVAL	/CLEAR SYMBOL VALUE
	DCA ZFLG	/CLEAR Z-BIT FLAG
	DCA INFO	/CLEAR INFORMATION FLAG
	DCA SIGN	/RESET SIGN TO "IOR"
	PUSHJ	/RECURSIVE CALL TO "CSYM"
	CSYM	/ADDRESS FOR PUSHJ HANDLER
	TAD PF	/WHICH PASS?
	SNA	/DON'T PUNCH ON PASS1 OR PASS3
	JMP MQUT	/PASS1, JUST RESET POINTERS
	SPA CLA	/PASS 3?
	JMP I POUTI	/YES: ASSEMBLY LISTING
	TAD INFO	/NO, PASS 2, ANYTHING TO PUNCH?
	SNA CLA	
	JMP MQUT	/NO, RESET FLAGS
	CLL	/CLEAR LINK FOR BINP ROUTINE
	TAD BINW	/GET ASSEMBLED BINARY WORD
	JMS I PUNI	/PUNCH IT
MQUT,	TAD INFO	/WAS THERE VALID INFORMATION ON THIS LINE?
	SZA CLA	
	ISZ AACT	/YES, INCREMENT CURRENT LOCATION COUNTER
	NOP	/PREPARE FOR ASSEMBLING INTO 7777
	JMP MAIN	/GO BACK TO MAIN LOOP
POUTI,	POUT	/ADDRESS OF PASS 3 HANDLER

/PAGE 10
 /MAIN CHARACTER PROCESSOR
 /GATHERS SYMBOLS ON A LINE-BY-LINE BASIS
 /SYMBOLS MUST BEGIN WITH A LETTER
 /LENGTH: 17 REGISTERS

CSYM, DCA SYM1 /CLEAR SYMBOL STORAGE WORDS
 DCA SYM2
 DCA SYM3
 JMS I AAA /GET A LEGAL CHARACTER
 JMS I NCHEKI /WHAT KIND OF CHARACTER IS IT?
 JMP I NUMBI /NUMBER = GO PROCESS
 ISZ INFO /LETTER, INDICATE VALID INFORMATION
 TAD M2 /GET UP COUNTERS = 2 CHARS PER WORD
 DCA TEM1
 TAD M4 /AND 3 WORDS PER SYMBOL
 DCA TEM2
 JMP I GETSP /GO PACK IT

 NXCH, JMS I AAA /GET A LEGAL CHARACTER
 JMS I NCHEKI /WHAT TYPE OF CHARACTER IS IT?
 JMP I GETSP /NUMBER LEGAL NOW
 JMP I GETSP /LETTER= GO PACK IT

 GETSP, GETS

/PAL PUNCH AND TYPE ROUTINE
 /LENGTH: LOPUN: 6 REGISTERS

LOPUN, 0
 TSF
 JMP .-1
 TLS
 CLA /EXIT WITH CLEARED AC
 JMP I LOPUN

/THE PAUSE PSEUDO-OP
 /LENGTH: 5 REGISTERS
 STOP, HLT CLA /HALT
 DCA SWITCH /RESET END-OF-TAPE SWITCH
 CMA /EMPTY READER
 DCA RCNT /BUFFER
 POPJ /EXIT

L02, /END OF PAGE

PAUSE